Newsgroup: it.comp.aiuto
Titolo: Sripts .vbs: chi mi aiuta?

Uso Thunderbird come programma di posta ed ho la necessità' di avere un
avviso "stile outlook" quando mi arriva nuova posta; non sempre sono
presente al computer e l'icona nella systray e l'avviso sonoro non mi basta
:-).
Ho trovato un'estensione che permette di utilizzare un "notificatore"
esterno che puo' essere un eseguibile oppure uno scripts.
Lo scripts in questione che mi piacerebbe utilizzare, visualizza un
assistente di Office (merlino) che ti "avverte" dell'arrivo di una nuova
mail, ma purtroppo dopo averti avvisato scompare.
E' possibile (per chi lo sa' fare) modificare questo scripts (e dove???) in
modo che sia possibile impostare il tempo di permanenza sullo schermo (tipo
2000000000000 secondi :-) ) oppure che rimanga visibile fino a che non ci
clicco sopra????
Sicuramente non e' il newsgroups piu' adatto, pero' in alternativa sapete
dove mi posso rivolgere?
Ciuaz
'Visual Basic -Display a notifier dialog box
Option Explicit
On Error Resume Next
'------------------------------------------------------------------------------
' Checks the number of arguments
'------------------------------------------------------------------------------
Sub CheckArguments(nb)
    If WScript.Arguments.Count <> nb Then
        WScript.Echo "Need "& nb & "parameters, received " & WScript.Arguments.Count
        WScript.Quit
    End If
End Sub
'------------------------------------------------------------------------------
' Displays the message with the MS agent
'------------------------------------------------------------------------------
Function agentNotify(message)
    Dim objAgentCtl
    Dim strPfad
    Dim objAgent
    Dim strCharacter
    strCharacter = "Merlin"
    agentNotify = false
    Set objAgentCtl = WScript.CreateObject("Agent.Control.1", "objAgent_")
    If IsObject(objAgentCtl) Then
        objAgentCtl.Connected = True
        strPfad = objShell.ExpandEnvironmentStrings("%WINDIR%\MSAgent\Chars\" &
strCharacter & ".acs")
        objAgentCtl.Characters.Load strCharacter, strPfad
        Set objAgent = objAgentCtl.Characters(strCharacter)
        If isObject(objAgent) Then
            message = message & " . . . . . . . . . . . ."
            With objAgent
                .Show
                .Speak displayedString
                .Hide
            End With
            agentNotify = true
            WScript.Sleep 10000
        End If
    End If
End Function
'------------------------------------------------------------------------------
' Checks the number of arguments
'------------------------------------------------------------------------------
Function dialogNotify(message)
    Dim intSecondsToWait
    Dim strTitle
    Dim intButton
    intSecondsToWait = 7
    strTitle = "New mail"
    intButton = objShell.Popup(message, intSecondsToWait, strTitle, 0)
    dialogNotify = true
End Function
'==============================================================================
'==============================================================================
' Main Program
'==============================================================================
'==============================================================================
Dim objShell
Dim subject
Dim sender
Dim recipients
Dim size
Dim folder
Dim acctype
Dim accname
Dim displayedString
Dim ret
CheckArguments 7
' Grab all of the arguments
subject = WScript.Arguments(0)
sender = WScript.Arguments(1)
recipients = WScript.Arguments(2)
size = WScript.Arguments(3)
folder = WScript.Arguments(4)
acctype = WScript.Arguments(5)
accname = WScript.Arguments(6)
displayedString = "From " & sender & ": " & vbCrLf & subject
Set objShell = WScript.CreateObject("WScript.Shell")
ret = agentNotify(displayedString)
If ret <> true Then
    ret = dialogNotify(displayedString)
End If
--------------------------------
Inviato via 
http://arianna.libero.it/usenet/

oppure che rimanga visibile fino a che non ci clicco sopra????
Per questo fammici lavorare un pò.
Ciao
Lorenzo

 Lorenzo


>ma non funziona :-(
OOOPS.
Non serve avere installato Thunderbird (per vari motivi non
posso) per provare, basta incollarlo in un file vbs e lanciarlo.
Ora sono sicuro:
ritorna al codice originale
sostituisci la riga
.Hide
con
.Wait
e modifica la riga
intSecondsToWait = 7
con
intSecondsToWait = 1200
dovrebbe aprirsi oltre che Merlino anche una popup
che rimane visibile per 20 minuti prima di autochiudersi
e portarsi dietro l'assistente. Prova così.
Ciao
Lorenzo


 GRANDE!!!!!! Adesso funziona.!!!!
Tra l'altro per chiudere il tutto si da' OK sulla popup.
Grazie!!
Ciauz
